
Overview 
The library described herein is designed for use with the Pascal programming language, specifically the Mad-
Pascal dialect with the intended target platform the Atari 8 bit home computer.


The library was initially written in Action! in 2015 and included only the base windowing system.  Over time it 
was expanded to include general purpose routines, and gadgets (which are windowing system add-ons).  The 
C version of the library is a conversion of the Action! library, done in 2022.  It includes all the windows and 
gadgets of the Action! library, and utilizes direct screen memory I/O like the Action! library.  There are a few 
functions that are either a) not yet converted; or b) not needed due to C having a similar function already.  The 
Pascal version presented here, is a conversion of the published 1.00 C version, done by Amarok (on 
AtariAge).  Many thanks for his hard work!


The major features offered by the library are:


• Window System

• The window management system allows the programmer to open and close windows with different styles.  

To reduce complexity and overhead it is a LIFO (last in first out) design.  It is intended for the programmer 
to keep track of the call stack.


• Gadgets

• Gadgets are windowing system add-ons which are designed to provide simple things like alert boxes, 

progress bars, and input controls.


• Menus

• Menus are a windowing system add-on which are designed to provide menu controls.


• Input/Output

• The input/output routines written in Action! were not converted because C has routines to do the same 

thing.  As such, they were note converted to Pascal either.


• String Manipulation

• Functions to aid with string manipulation and character conversion.


• DOS Functions

• The DOS routines written in Action! were not converted because the CC65 Atari library has routines to do 

the same thing.  As such, they were not converted to Pascal either.


• Miscellaneous

• Helper functions that don’t fall into any particular category, including waiting with and without keystrokes.  

The Action! Wait() function was not converted because CC65 provides sleep() which does the same thing.  
As such, Wait() was not converted to Pascal either.


References in this documentation that refer to void are meant to represent no value.

References in this documentation that refer to Byte are meant to represent an integer value from 0 to 255.

References in this documentation that refer to Word are meant to represent an integer value from 0 to 65,535.

Local Variables Warning 
When programming functions with a large number of variables, or a few variables of a large size, you may run 
into compilation errors that state there are “Too many local variables”.  This is because local variables are 
normally placed on the stack which has room for 256 bytes.


You can overcome this by using the compiler command line option “-static-locals” which will move local 
variables to the data segment and off the stack.  However doing it via the CLI applies to all the code.


The better way to overcome this is by using compiler directive #pragma static-locals before and after 
the function in the source code.  This will isolate moving the locals to the data segment to just that function.


You can see an example of the usage in the libraries example program appdemo.c surrounding the 
FormInput() function.


Example usage:

#pragma static-locals(push, on)
byte FormInput(void) {
...
}
#pragma static-locals(pop)

Push will change the setting and save the current setting.  Pop restores the saved setting.

Code Management 
Size 
When using all of the components of the library, the code size of your application could start to become rather 
large.  If you find your program no longer fits in available memory or does not have enough memory for 
variables after loading, you may need to optimize the compile environment.


To optimize the code size, copy all of the library files to your project directory.  Subsequently modify your 
applications source files to include the library files from this location rather than the original library source 
location. 


Now that your application is including the library from your application project location, you can proceed.  You 
will want to cross reference functions defined by the library with those your application uses, including 
dependencies of the library functions (some library functions call others).  This documentation will help 
identify the dependencies.


One you have identified all of the library functions (and their dependents) used by your application,  you will 
then modify the library files (in your project directory, NOT the original source!).  In these library files, you 
will remove any functions that are not needed by your application, thus reducing the overall compiled code 
size.


Compilation 
In general when compiling a project with Mad-Pascal, the easiest method is as follows.  This may not suit a 
complex project which could require the use of make or a build script. 


Typically the calls will be:

mp.exe src\program.pas -ipath:src
mads.exe src\program.a65 -x -i:<MadPascalPath>\base -o:bin\program.xex

Parameters:

src\program.pas is the name of the program source file.


-ipath:src is the location of the programs source files


src\program.a65 is the name of the program source file.


-x means ???.


-i: is the location of the Mad-Pascal base files.


-o: tells Mad-Pscal the name to use for the executable.

Organization 
You can pull the library components in as needed, though there is some loose required order due to the library 
using parts of itself.


First you will want to pull in any Mad-Pascal headers that are needed.  The following are the base ones used 
when using all parts of the C Library:

Crt
SysUtils

Then you will want to pull in the Mas-Pascal Window Library headers that are needed in this order.  All of the 
library sources require “a8defines.pas”.  Only “a8libwin.pas” requires “a8defwin.pas”.  It won’t hurt 
in include these more than once as they establish a definition they have been loaded and only load if the 
definition is not present.

#include "a8defines.pas"
#include "a8defwin.pas"

Last you will pull in the the Mad-Pascal Window Library sources that are needed in this order:

#include "a8libmisc.pas"
#include "a8libstr.pas"
#include "a8libwin.pas"
#include "a8libgadg.pas"
#include "a8libmenu.pas"

Requirements 
Mad-Pascal Standard Library Dependency 
This library depends upon some of the Mad-Pascal standard library functions.  Library routines will list their 
standard function dependencies in the API reference which follows in this documentation.


Routines from the Mad-Pascal standard library that are needed:

Char()
ClrScr
Dec()
DPeek()
FillChar()
Inc()
Length()
Move()
Peek()
Pointer()
Poke()
SetLength()
Space()

Symbol Table 
The symbols used by this library are as follows.  Many of the names are re-used throughout the library, and 
are kept to a short length to converse space.  


bC
bD
bE
bF
bH
bI
bK
bL
bM
bN
bP
bR
bS

bT
bU
bW
bZ
cL
cR
cS
iL
pA
pS
h
w
x

xp
y
yp

baW
baWM
cpWn
vCur

Result
tmpStr  
File Reference 

A8DEFINES.PAS
All definitions used throughout the library.  


This should be included FIRST at the top of the main program file, and should be included in any program 
that uses the library routines.


A8DEFWIN.PAS
Window type definitions and variables used by the window system portion of the library.  


If the windowing system is used, this file should be included immediately after A8DEFINES.PAS, and 
BEFORE A8LIBWIN.PAS.


A8LIBGADG.PAS
Collection of gadgets (add-ons) for the window system.


When using these routines, A8LIBWIN.PAS MUST be included before.


A8LIBMENU.PAS
Collection of menu routines which simplifies program navigation.


When using these routines, A8LIBWIN.PAS MUST be included before.


A8LIBMISC.PAS
Collection of routines that don’t fall into the other categories, such as waiting for key.


A8LIBSTR.PAS
Collection of string manipulation routines that augment the Mad-Pascal standard library routines and which 
are specific to the Atari platform.

A8LIBWIN.PAS
Collection of window routines that make up the text window system.


When using these routines, A8DEFINES.PAS, A8DEFWIN.PAS, and A8LIBSTR.PAS MUST be included 
before.

API Reference 
Window System (A8LIBWIN.PAS) 

void WBack(bN: Byte)

Parameters:	 bN = Internal code of character

Returns:	 void

Requires:	 A8DEFWIN.PAS

Standard     - DPeek(), FillChar(), Pointer()

Description 
Sets the background “image” that covers the entire screen.  This is a single character to repeat in every cell.


Using large footprint characters (a lot of pixels) can make the program elements like windows and menus 
harder to see.  It is best used with small footprint characters like the ‘.’.  With a custom character set, this 
function could be advantageously used.


Byte WClose(bN: Byte)

Parameters:	 bN = Window handle number

Returns:	 Byte = Success status

	 	 	  0 = Successful

	 	 	  WENOPN = Window not open (default)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
Standard     - DPeek(), FillChar(), Inc(), Move(), Pointer()

Description 
Closes an open window specified by the handle bN.  


If window is not open, no action is taken.


It is up to the programmer to close windows in the proper order - the last one opened should be the first one 
closed.  If an earlier window is closed before a more recent overlapping window, the screen contents will not 
be reflected accurately when the latter is closed (it will show remnants of the earlier window).

Byte WClr(bN: Byte)

*** Currently NOT Implemented *** 

Parameters:	 bN = Window handle number

Returns:	 Byte = Success status

	 	 	 0 = Succesful

	 	 	 WENOPN = Window not open (default)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
A8LIBSTR.PAS - StrInv()
Runtime      - DPeek(), Move(), Pointer()

Description 
Clears the contents of the window referenced by window handle bN.  Effectively clearing the screen of the 
windows interior dimensions (excluding frame).

Byte WDiv(bN, y, bD: Byte)

*** Currently NOT Implemented *** 

Parameters:	 bN = Window handle number

	 	 y = Window row to display divider

	 	 bD = On/Off flag

	 	 	 WON = On (show divider)

	 	 	 WOFF = Off (remove divider)

Returns:	 Byte = Success status

	 	 	 0 = Succesful

	 	 	 WENOPN = Window not open (default)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
Standard     - DPeek(), Move(), Pointer()

Description 
Draws a divider line in the window referenced by handle bN.  


The divider is drawn on row y of the window.


The bD (display on/off) parameter is passed as WON, the bar will 
be displayed.  With WOFF, the bar will be removed which will 
blank the contents on the window row and restore the window 
frame.


Calling WDiv() with WOFF is also a quick way to clear one line of 
a window. 
void WInit(void)

Parameters:	 void
Returns:	 void
Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS 
Standard     - ClrScr, FillChar(), Poke()

Description 
Used to initialize the window management system.  It should be called before any other windowing system 
call.


In addition to defaulting all the windowing system variables, it will perform the following:

• Turn the cursor off (poke 752, 1)

• Set the left screen margin to 0 (poke 82, 0)

• Set the cursor position to the top left corner (0,0)

• Clear the screen


The library is built to handle 10 windows.  You can alter this routine for more or less as your program requires.  
Memory requirements will increase or decrease as the number is changed.  Increasing the number may also 
necessitate increasing the window system storage space by increasing the value of WBUFSZ in file 
A8DEFWIN.PAS.

Byte WOpen(x, y, w, h, bT: Byte)

Parameters:	 x = Column of screen for left edge of window

	 	 y = Row of screen for top edge of window

	 	 w = Width of window in columns

	 	 h = Height of window in rows

	 	 bT = Inverse video flag (optional)

	 	 	 WON = Inverse video

	 	 	 WOFF = Normal video (default)

Returns:	 Byte = Window handle number

	 	 	 -or-

	 	 	 WENONE = No window handles available

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
Standard     - DPeek(), FillChar(), Inc(), Move(), Pointer()

Description 
Opens a window on the screen with a single line border.  The screen contents under the window are saved, 
then restored when the window is closed.


Top left coordinate is specified by x and y.  The width and height are specified with w and h.  If the inverse 
flag, bT, is set, the window is drawn and filled in inverse video.

Byte WOrn(bN, bT, bL: Byte; pS: string)

Parameters:	 bN = Window handle number

	 	 bT = Top or bottom of window designation

	 	 	 WPTOP = Top border

	 	 	 WPBOT = Bottom border

	 	 bL = Left, right, or center of window designation

	 	 	 WPCNT = Center

	 	 	 WPLFT = Left side

	 	 	 WPRGT = Right side

	 	 pS = Pointer to character string of title text

	 	 	 Maximum size is 36 characters!

Returns:	 Byte = Success status

	 	 	 0 = Succesful

	 	 	 WENOPN = Window not open (default)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
A8LIBSTR.PAS - StrAI(), StrInv()
Standard     - DPeek(), Inc(), Length(), Move(), SetLength()

Description 
Sets a window ornament to text string s with decorations on the window referenced by bN, on either the top 
or bottom border as given by bT, and left or right side as given by bL.


If an ornament is to be set, the window itself must be large enough to accommodate it, along with any other 
assigned ornaments.  For a single ornament, a minimum window width should be the title length plus four 
characters (two characters for the ornaments on either side of the tile, and two characters for the window 
frame where the ornaments can’t be drawn).  Because of this, the maximum length of a title is 36 characters.


If multiple ornaments are used on top or bottom at the same time, care must be taken to ensure the window 
size is large enough, or the ornament size is small enough, to accommodate both ornaments.

Byte WPos(bN, x, y: Byte)

*** Possibly not working *** 

Parameters:	 bN = Window handle number

	 	         -or-

	 	         WPABS = Absolute screen position

	 	 x = Window column to move cursor to

	 	 y = Window row to move cursor to

Returns:	 Byte = Success status

	 	 	 0 = Succesful

	 	 	 WENOPN = Window not open

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
Standard     - 

Description 
Moves the window systems virtual cursor to the screen position of the specified x and y coordinates within 
the window referenced by window handle bN.  

Byte WPrint(bN, x, y, bI: Byte; pS: string)

Parameters:	 bN = Window handle number

	 	 x = Window column to print text

	 	 y = Window row to print text

	 	 bI = Inverse flag 

	 	 	 WON for inverse

	 	 	 WOFF for normal

	 	 pS = Pointer to character string of text to print

	 	 	 Maximum size is 38 characters!

Returns:	 Byte = Success status

	 	 	 0 = Succesful

	 	 	 WENOPN = Window not open (default)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
A8LIBSTR.PAS - StrAI(), StrInv()
Standard     - DPeek(), Inc(), Length(), Move(), SetLength()

Description 
Prints text string pointed to by pS at the virtual cursor position of x and y within the window referenced by 
window handle bN.  You can force inverse video text by specifying WON in the bI parameter, otherwise use 
WOFF.


A minimum window width should be the text length plus two characters (for the window frame).  Because of 
this, the maximum length of a text is 38 characters.

Byte WPut(bN: Byte; x: Char)

Parameters:	 bN = Window handle number

	 	 x = Character to put

Returns:	 Byte = Success status

	 	 	 0 = Succesful

	 	 	 WENOPN = Window not open

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
Standard     - Byte(), Char(), DPeek(), Inc(), Poke()

Description 
Outputs the character specified by x at the window systems virtual cursor within the window referenced by 
window handle bN.  


Increments the window systems virtual cursor by one column.


If the window was created with the inverse flag set, the character will be inversed to match.


Byte WStat(bN: Byte)

Parameters:	 bN = Window handle number

Returns:	 Byte = Window status

	 	 	 WON = In use (window ON)

	 	 	 WOFF = Not in use (window OFF)

Requires:	 A8DEFWIN.PAS

Description 
Returns the status of the window specified by the handle bN.  

Gadgets (A8LIBGADG.PAS) 

void GAlert(pS: string)

Parameters:	 pS = Pointer to character string to display

	 	 	 Maximum size is 38 characters!

Returns:	 void

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
	 	 A8LIBWIN.PAS  - WOpen(), WOrn(), WPrint(), WClose()

A8LIBMISC.PAS - WaitKCX()
Standard    - Length()

Description 
Displays a screen centered modal window with the title “Alert” 
and the message text of the string pointed to by char pointer pS. 
It will display an OK “button” beneath the text and wait for 
keystroke, which will be consumed.


Calling GAlert will consume one window handle while it is open.


Because the window will have a frame, the maximum message 
length is 38 characters.

Byte GButton(bN, x, y, bD, bS: Byte; pA: TStringArray)

Parameters:	 bN = Window handle number

	 	 x = Window column to start get

	 	 y = Window row

	 	 bD = Initial selected button

	 	 	 GDISP = Display only and exit

	 	 bS = Number of buttons in array pA

	 	 pA = Pointer to ragged array of button name strings

Returns:	 Byte = Button number selected or exit status

	 	 	 XESC = Exited with ESCape (no button chosen)

	 	 	 XTAB = Exited with TAB (no button chosen)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
	 	 A8LIBWIN.PAS  - WPrint()

A8LIBMISC.PAS - WaitKCX()
Standard    - Dec(), Inc(), Length()

Description 
Displays a row of buttons and gets selection from user.


If the initial selection indicator (bD) is passed as GDISP, then the 
buttons will be displayed and the function will exit (none will be 
highlighted).


It is up to the programmer to define the button ornaments, if any.  
For example: [ OK ].  In this example the [ and ] are the 
ornaments enclosing the 4 character string space OK space.  The 
entire string will be inversed when selected, including the 
ornaments.


Care must be taken on the total length of the button strings 
contained in ragged array pointer pA.  The total should be no more 
than 38 for a window that is 40 wide.


Keys accepted are:

LEFT\+ = Move button selector left

RIGHT\* = Move button selector right

UP\- = Move button selector left

DOWN\= = Move button selector right

ESCAPE	 	 = Exits without selection (returns XESC)

TAB	 	 	 = Exits without selection (returns XTAB)

ENTER		 	 = Accepts current selected button and exits (returns selected button #)

Byte GCheck(bN, x, y, bI, bD: Byte)

Parameters:	 bN = Window handle number

	 	 x = Window column to start get

	 	 y = Window row

	 	 bI = Display Only indicator

	 	 	 GDISP = Display only and exit

	 	 	 GEDIT = Edit checkbox

	 	 bD = Default initial value

	 	 	 GCON = Checked

	 	 	 GCOFF = Unchecked

Returns:	 Byte = Checked status or exit status

	 	 	 GCON = Checked

	 	 	 GCOFF = Not Checked

	 	 	 XESC = Exited with ESCape (no value chosen)

	 	 	 XTAB = Exited with TAB (no value chosen)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
	 	 A8LIBWIN.PAS   - WPrint()

A8LIBMISC.PAS  - WaitKCX()

Description 
Displays a checkbox ( [ ] ) and gets selection from user.  


Unlike many other input gadgets, the text for the option is not 
included and should be displayed separately in the window using 
WPrint() prior to calling GCheck().


When the checkbox is marked, an inverse video X will be 
displayed, otherwise it will be an inverse space.  When the 
function exits, the set value will be displayed in normal video.  
ENTER must be used to set (lock) the value to be returned, 
otherwise the default value passed in is re-displayed.


If the display only indicator (bI) is passed as GDISP, then the 
checkbox will be displayed and the function will exit.  Display Only will respect default values and represent 
them accordingly.  This is useful for drawing the checkbox on a form before selection is to occur.


Keys accepted are:

ESCAPE	 	 = Exits without selection (returns XESC)

TAB	 	 	 = Exits without selection (returns XTAB)

SPACE 	 	 = Toggle value of checkbox (display only)

X/x	 	 	 = Acts just like SPACE

ENTER		 	 = Accepts (sets and locks to displayed current value) and exits

Byte GInput(bN, x, y, bT, bS: Byte; var pS: string)

Parameters:	 bN = Window handle number

	 	 x = Window column to start get

	 	 y = Window row

	 	 bT = Allowed character type

	 	 	 GANY = Any non-cursor control character

	 	 	 GALNUM = Any Alpha-Numeric character (0-9, a-z, A-Z, <space>)

	 	 	 GALPHA = Alphabetic characters only (a-z, A-Z, <space>)

	 	 	 GNUMER = Numeric characters only (0-9, ., -)

	 	 bS = Display size for string (max 38)

	 	 pS = Pointer to text string to input/edit (max 128)

Returns:	 Byte = Success indicator

	 	 	 TRUE = String was modified

	 	 	 FALSE = String was not modified

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
A8LIBMISC.PAS - WaitKCX(), IKC2ATA()

	 	 A8LIBWIN.PAS  - WPrint()
Standard      - Byte(), Char(), Dec(), FillChar(), Inc(), Length(),

     Move(), SetLength()

Description 
Edits a large string in a smaller display window by scrolling 
through the string and displaying only a portion at a time, much 
like modern operating system input fields.  


The edit area is opened in the window handle referenced by bN.  
The edit area is placed at the x and y position in the window.  
The maximum size of the edit area is specified by bS, and the 
maximum should be considered to be 38 (given a window that is 
40 characters wide).


The initial edit area contents will be a copy of the string passed 
as pS.


If the input is exited using ESC, the string passed will be left in tact.  If the input is exited using the ENTER key, 
any edits made will be copied to the string passed via pointer pS.  This means you can not pass a static text 
string such as “Hello World”, it MUST be a string pointer.

Keys accepted are:

LEFT\+ = Move cursor left

RIGHT\* = Move cursor right

DEL = Delete character left of cursor (or 1st char if cursor is at position 1)

Control-DEL	 = Delete character at cursor (move remainder left 1 position, add space at end)

Shift-DEL = Delete entire string contents (moves cursor to position 1 of text string)

INSERT	 	 = Insert space at cursor (character at end of text string will be lost)

Control-Shift-S = Move cursor to beginning of string

Control-Shift-E = Move cursor to end of string

ESCAPE	 	 = Cancel edits and exit

ENTER		 	 = Accept edits and exit

void GProg(bN, x, y, bS: Byte)

Parameters:	 bN = Window handle number

	 	 x = Window column to display bar at

	 	 y = Window row to display bar at

	 	 bS = Bar size (Percent complete), Range 0 to 100.

Returns:	 void

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
	 	 A8LIBWIN.PAS  - WPrint()

Standard      - FillChar(), Space()

Description 
Displays a progress bar at the x and y position within the window 
referenced by window handle bN.  The percentage complete is 
referenced by bS, and has a range of 0 to 100.   The bar will 
always be displayed with 20 characters of length, so the window 
it is placed in should be a minimum of 22 wide to account for the 
frames.

Byte GRadio(bN, x, y, bD, bE, bI, bS: Byte; pS: TStringArray)

Parameters:	 bN = Window handle number

	 	 x = Window column to start get

	 	 y = Window row

	 	 bD = Direction of button placement

	 	 	 GHORZ = Horizontal (side by side)

	 	 	 GVERT = Vertical (stacked)

	 	 bE = Display Only indicator

	 	 	 GDISP = Display only and exit 

	 	 	 GEDIT = Edit

	 	 bI = Initial selected button

	 	 bS = Number of buttons in array pS

	 	 pS = Pointer to ragged array of button name strings

Returns:	 Byte = Button number selected or exit status

	 	 	 XESC = Exited with ESCape (no button chosen)

	 	 	 XTAB = Exited with TAB (no button chosen)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
	 	 A8LIBWIN.PAS  - WPrint(), WPos(), WPut()

A8LIBMISC.PAS - WaitKCX()
Standard      - Dec(), Inc(), Length()

Description 
Displays a selection of radio buttons and gets a selection of one 
from user.  


Only one button from the defined group can be selected. When 
there is a need for multiple option selection GCheck() should be 
used instead.


The buttons will be arranged in the direction specified by bD.  
Valid directions are GHORZ or horizontal (side by side), or GVERT 
for vertical (stacked) alignment.  Care should be taken to ensure 
the window boundaries are large enough to accommodate the 
buttons, especially when aligning horizontally.  For horizontal 
buttons, it is only reasonably to expect 3 or 4 buttons to fit in the 
38 columns available inside a window frame.  Each horizontal button is separated by 2 spaces.  For this 
reason, it is recommended to use vertical alignment (GVERT) to stack the buttons for more than 3 buttons.


If the initial selection indicator (bI) is passed as GDISP, then the buttons will be displayed and the function 
will exit (none will be highlighted).  This is useful for drawing the buttons on a form before selection is to 
occur.

Keys accepted are:

LEFT\+ = Move button selector left

RIGHT\* = Move button selector right

UP\- = Move button selector left

DOWN\= = Move button selector right

ESCAPE	 	 = Exits without selection (returns XESC)

TAB	 	 	 = Exits without selection (returns XTAB)

SPACE 	 	 = Set currently selected button as choice

ENTER		 	 = Accepts current selected button and exits (returns selected button #)

Byte GSpin(bN, x, y, bL, bM, bI, bE: Byte)

Parameters:	 bN = Window handle number

	 	 x = Window column to display value at

	 	 y = Window row to display value at

	 	 bL = Lowest allowed value

	 	 bM = Maximum allowed value

	 	 bI = Initial value

	 	 bE = Display only indicator

	 	 	 GDISP = Display only and exit

	 	 	 GEDIT = Edit

Returns:	 Byte = Value selected or exit status

	 	 	 XESC = Exited with ESCape (no value chosen)

	 	 	 XTAB = Exited with TAB (no value chosen)

Requires:	 A8DEFINES.PAS

A8DEFWIN.PAS
	 	 A8LIBWIN.PAS  - WPrint()

A8LIBMISC.PAS - WaitKCX()
A8LIBSTR.PAS  - ByteToStr3()
Standard      - Dec(), Inc()

Description 
Displays value bP, considered the starting/default value, and 
allows value change via spinner controls.  The lowest value is 
limited to bL.  The maximum value is limited to bM.  


Any byte value is allowed for the limits and default value.  The 
upper limit can be up to 250, and is limited by the function at this 
value.  This is because the input gadgets use the the 253, 254, 
and 255 as specific return values that indicate how the gadget 
was exited.  The spinner gadget is an exception in that it is a 
hybrid.  It will return those values, and it returns the selected 
value.  Realistically, the foreseen use case is from 0 to 100.


Keys accepted are:

LEFT\+ = Decrease value

RIGHT\* = Increase value

UP\- = Increase value

DOWN\= = Decrease value

ESCAPE	 	 = Exits without setting value (returns XESC)

TAB	 	 	 = Exits without setting value (returns XTAB)

ENTER		 	 = Accepts current value and exits (returns value)

Menus (A8LIBMENU.PAS) 

Byte MenuV(bN, x, y, bI, bS, bC: Byte; pS: TStringArray)

Parameters:	 bN = Window handle number

	 	 x = Window column to display menu at

	 	 y = Window row to display menu at

	 	 bI = Inverse selection on exit flag

	 	 	 WON = Leave menu selection in inverse video

	 	 	 WOFF = Return menu selection to normal video

	 	 bS = Start item selection number

	 	 bC = Number of menu items in array pS

	 	 pS = Pointer to array of strings containing menu items

Returns:	 Byte = Number of item chosen or exit value

	 	 	 XESC = Exited with ESCape (no item chosen)

	 	 	 XTAB = Exited with TAB (no item chosen)


Requires:	 A8DEFINES.PAS
A8DEFWIN.PAS

	 	 A8LIBWIN.PAS  - WPrint()
A8LIBMISC.PAS - WaitKCX()
Standard      - Dec(), Inc(), Length(), Move(), SetLength()

Description 
Displays a list of menu items at the x and y coordinates within 
the window referenced by window handle bN. 


The currently selected menu item will be highlighted (displayed in 
inverse video), while the remaining items will be in normal video.  


The menu will have the following navigation key controls:

UP/-	 	 = Move cursor (selection) up

DOWN/=	 = Move cursor (selection) down

LEFT/+	 = Move cursor (selection) up

RIGHT/*	 = Move cursor (selection) down

ENTER		 = Accept selected item

ESCAPE 	 = Abandon selection and return

TAB 	 	 = Abandon selection and return


The initially selected item will be the one referenced by bD.


If the selector scrolls past the bottom it will be returned to the top.  Likewise if the selector scrolls past the top 
it will set to the bottom.


If the inverse on exit parameter (bI) is set to WON, the currently highlighted (selected) menu item will remain in 
inverse video at exit.  This is useful if you have sub-menus and want to see the “breadcrumbs” of previous 
selections.

If the inverse on exit parameter (bI) is set to WOFF, the currently highlighted (selected) menu item will be re-
displayed in normal video at exit.  This is useful for generating input forms and using MenuV() as a field 
selector.


The number of the item selected will be returned once a selection is accepted.

If ESCAPE is used to exit the menu, it will return 0 (XESC).

If TAB is used to exit the menu, it will return 99 (XTAB).

String Manipulation (A8LIBSTR.PAS) 

string ByteToStr3(bN: Byte)

Parameters:	 bN = byte to convert

Returns:	 string = converted byte in string form

Requires:	 n/a


Description 
Converts single byte referenced by bC to a 3 character string.


This is generally useful for converting numbers to a padded fixed length string for display or reporting 
purposes.


Byte CharAI(bC: Byte)

Parameters:	 bC = byte to convert

Returns:	 Byte = converted byte

Requires:	 n/a


Description 
Converts single byte referenced by bC from the ATASCII code representation to the internal code 
representation.


This is generally useful for putting characters or copying text strings directly to screen memory.


void StrAI(pS: PByte; bS: Byte)

Parameters:	 pS = Pointer to text string

	 	 bS = Size (number) of chars in string to convert

Returns:	 void

Requires:	 n/a


Description 
Converts elements (characters) of the string referenced by pS from the ATASCII code representation to 
the internal code representation up to size bS bytes in length.


This is generally useful for putting characters or copying text strings directly to screen memory.


This is the opposite of StrIA().

void StrIA(pS: PByte; bS: Byte)

*** Currently NOT Implemented *** 

Parameters:	 pS = Pointer to text string

	 	 bS = Size (number) of chars in string to convert

Returns:	 void

Requires:	 n/a


Description 
Converts elements (characters) of the string referenced by pS from the internal code representation to the 
ATASCII code representation up to size bS bytes in length. 


This is the opposite of StrAI().


void StrInv(pS: PByte; bS: Byte)

Parameters:	 pS = Pointer to text string

	 	 bS = Size (number) of chars in string to inverse

Returns:	 void

Requires:	 n/a


Description 
Inverses (inverse video) elements (characters) of the string referenced by pS up to size bS bytes in length.

Miscellaneous (A8LIBMISC.PAS) 

Byte IKC2ATA(bN: Byte)

Parameters:	 bN = Internal key code

Returns:	 Byte = ATASCII character code

	 	 	 Or unconverted internal code (see Description)

	 	 	 Or KNOMAP (199) for internal codes with no character mapping (see Description)

Requires:	 A8DEFINES.PAS

Description 
Converts internal key code to ATASCII character code.


Performs conversion for all internal key codes with value less than 192.  If the internal code passed in is 
greater than 191, it is returned unmodified.  If the internal code passed in is greater than 127 and does not 
have a character mapping, KNOMAP (199) is returned.  Key code 199 is not bound to any keystroke 
combination.


Byte WaitYN(bD: Byte)

*** Currently NOT Implemented *** 

Parameters:	 bD = Flag for display of ? prompt

	 	 	 WON = Display ? prompt

	 	 	 WOFF = Do not display ? prompt

Returns:	 Byte = 1 for Y or y pressed

	 	 	  0 for  N or n pressed

Requires:	 A8DEFINES.PAS

Standard       - Peek(), Poke()

Description 
Waits for a Y or N keystroke.  Upper and lower case letters are accepted.  Will optionally display a ‘?’ 
character at the current virtual window system cursor location if bD is set to WON.


The keypress is consumed before returning. 

30



Word WaitKCX(bI: Byte)

Parameters:	 bI = Flag to execute inverse function or not

	 	 	 WON = Yes

	 	 	 WOFF = No

Returns:	 Word = key code value of key pressed

	 	 	 KFHLP = Help key

	 	 	 KF1 = F1 key (1200XL/1400XL/1450XLD only)

	 	 	 KF2 = F2 key (1200XL/1400XL/1450XLD only)

	 	 	 KF3 = F3 key (1200XL/1400XL/1450XLD only)

	 	 	 KF4 = F4 key (1200XL/1400XL/1450XLD only)

	 	 	 KCAP = Caps key

	 	 	 KINV = Inverse key

	 	 	 KCOPT = Option key

	 	 	 KCSEL = Select key

	 	 	 KCSTA = Start key

Requires:	 A8DEFINES.PAS

Standard       - DPeek(), Peek(), Poke()

Description 
Waits for any keystroke, function key, or console key press.  Function keys include HELP, and F1 through F4.  
This function will process transient keys Caps and Inverse as well as returning the key stroke value.  This 
means caps-lock will be toggled on and off as the key is pressed.


The transient inverse keystroke will be toggled only if bI is passed as WON.


The keypress is consumed or debounced as necessary before the function returns.


This is intended for use on XL/XE computers. 

Usage Examples 
Stub Programs 

Stub Window
This demonstrates the very basics of the window system.  It shows how to include the library and open a 
window.


File: STUBWIN.PAS

// ------------------------------------------------------------
// Program: stubwin.c
// Desc...: A8 Library Stub Window Program
// Author.: Wade Ripkowski, amarok
// Date...: 20220914
// License: GNU General Public License v3.0
// Notes..: mp.exe src\stubwin.pas -ipath:src
//          mads.exe src\stubwin.a65 -x -i:<MadPascalPath>\base -o:bin\stubwin.xex
// ------------------------------------------------------------

// Pull in include files
uses
    a8defines, a8defwin, a8libwin, a8libmisc;

var

    // Variables
    bW1: Byte;
  
begin
    // Init Window System
    WInit;

    // Open window
    bW1 := WOpen(8, 5, 24, 9, WOFF);
    WOrn(bW1, WPTOP, WPLFT, 'Stub');
    WPrint(bW1, 5, 2, WOFF, 'Inverse');
    WPrint(bW1, 12, 2, WON, 'ATASCII');
    WPrint(bW1, WPCNT, 4, WOFF, 'Unfinished Bitness');
    WPrint(bW1, WPCNT, 6, WON, ' Ok ');

    // Wait for a keystroke or console key
    WaitKCX(WOFF);

    // Close window
    WClose(bW1);
end.

Stub Application Shell
This demonstrates a shell application using the window system.  It shows how to include the library and build 
the foundation of a larger application.


File: STUBAPP.PAS

// ------------------------------------------------------------
// Program: stubapp.pas
// Desc...: A8 Library Stub Application
// Author.: Wade Ripkowski, amarok
// Date...: 20220914
// License: GNU General Public License v3.0
// Notes..: mp.exe src\stubapp.pas -ipath:src
//          mads.exe src\stubapp.a65 -x -i:<MadPascalPath>\base -o:bin\stubapp.xex
// ------------------------------------------------------------

// Pull in include files
uses
    a8defines, a8defwin, a8libmisc, a8libstr, a8libwin, a8libgadg, a8libmenu;

// ------------------------------------------------------------
// Func...: About
// Desc...: About Dialog
// ------------------------------------------------------------
procedure About;
var
    bW1: Byte;

begin
    // Show window
    bW1 := WOpen(1, 6, 38, 15, WOFF);
    WOrn(bW1, WPTOP, WPLFT, 'About');
    WPrint(bW1, WPCNT,  1, WOFF, 'Demo Application');
    WPrint(bW1, WPCNT,  2, WOFF, 'Version 1.00-PAS');
    WPrint(bW1, WPCNT,  3, WOFF, '(C) 2022  Wade Ripkowski, amarok');
    WPrint(bW1, WPCNT,  5, WOFF, 'Application to demonstrate');
    WPrint(bW1, WPCNT,  6, WOFF, 'the MadPascal library.');
    WPrint(bW1, WPCNT,  7, WOFF, 'Converted from C.');
    WPrint(bW1, 4,      9, WOFF, 'V1-2021-Atari8: Action!');
    WPrint(bW1, 2,     10, WOFF, 'V1-C-2022-Atari8: C (CC65)');
    WPrint(bW1, 2,     11, WOFF, 'V1-PAS-2022-Atari8: PAS (MadPascal)');
    WPrint(bW1, WPCNT, 13, WON,  ' Ok ');

    // Wait for key
    WaitKCX(WOFF);

    // Close window
    WClose(bW1);
end;

// ------------------------------------------------------------
// Func...: SubMenu3
// Desc...: Sub menu routine
// ------------------------------------------------------------
procedure SubMenu3;
var
    bW1, bC: Byte;
    bD: Boolean;
const
    pcM: array[0..2] of string = (' Sub-Item 1 ', ' Sub-Item 2 ', ' Sub-Item 3 ');
begin
    bD := false;

    // Open window
    bW1 := WOpen(16, 10, 14, 5, WOFF);
    WOrn(bW1, WPTOP, WPLFT, 'Sub-Menu');

    // Loop until exit
    while not bD do
    begin
        // Display menu and get choice
        bC := MenuV(bW1, 1, 1, WOFF, 1, 3, pcM);

        // Process choice
        case bC of
            1: GAlert(' Sub-Item 1 selected. ');
            2: GAlert(' Sub-Item 2 selected. ');
            3: GAlert(' Sub-Item 3 selected. ');
            XESC: bD := true;
        end;
    end;

    // Close window

    WClose(bW1);
end;

var
    // Variables
    bW1, bW2, bC: Byte;
    bD: Boolean;
const
    pcM: array[0..4] of string =
      ( ' Sub-Menu 1 ', ' Sub-Menu 2 ', ' Sub-Menu 3 ', ' About      ', ' Exit       ');
begin
    bD := false;

    // Setup screen
    WInit;

    // Set Background
    WBack(14);

    // Open header window
    bW1 := WOpen(0, 0, 40, 3, WON);
    WPrint(bW1, WPCNT, 1, WOFF, 'A P P L I C A T I O N');

    // Open menu window
    bW2 := WOpen(13, 7, 12, 9, WOFF);
    WOrn(bW2, WPTOP, WPCNT, 'Menu');

    // Loop until done (Exit selected)
    while not bD do
    begin
        // Call menu
        bC := MenuV(bW2, 1, 2, WOFF, 1, 5, pcM);

        // Process choice
        case bC of
            1: GAlert(' Sub-Menu 1 selected. ');
            2: GAlert(' Sub-Menu 2 selected. ');
            3: SubMenu3;
            4: About;
            5: bD := true;
        end;

        // Exit on ESC as well
        if bC = XESC then
        begin
            bD := true;
        end;
    end;

    // Close windows
    WClose(bW2);
    WClose(bW1);
end.

Demo Program 

Demonstration Application
This demonstrates a fully functioning application using the window system, and several gadgets.  It shows 
how to include the library and build the foundation of a larger application.


File: APPDEMO.PAS

// ------------------------------------------------------------
// Program: appdemo.pas
// Desc...: A8 Library Demo Application
// Author.: Wade Ripkowski, amarok
// Date...: 20220914
// License: GNU General Public License v3.0
// Notes..: mp.exe src\appdemo.pas -ipath:src
//          mads.exe src\appdemo.a65 -x -i:<MadPascalPath>\base -o:bin\appdemo.xex
// ------------------------------------------------------------

// Pull in include files
uses
    Crt, a8defines, a8defwin, a8libmisc, a8libstr, a8libwin, a8libgadg, a8libmenu;

// ------------------------------------------------------------
// Func...: FormInput: Boolean
// Desc...: Demo use of input gadgets
// Returns: TRUE if accepted, else FALSE

// Notes..: Maximum local variable stack space is 256 bytes.
//          MUST use pragma static-locals to move variables to
//          BSS segment due to total size in this function.
// ------------------------------------------------------------
function FormInput: Boolean;
var
    bRA, bRB, bChap, bChbp, bChcp, bV: Byte;
    bW1, bM, bA, bB, bC, bD, bVp, bRAp, bRBp, bCha, bChb, bChc: Byte;
    
    // Input strings & navigation strings
    cA, cB, cC, cD: string[41];
    cF, cI, cR, cX: string[15];
const
    // Regular buttons, radio buttons, and data field names
    paB: array[0..1] of string = ('[ Ok ]', '[Cancel]');
    prA: array[0..2] of string = ('One', 'Two', 'Three');
    prB: array[0..2] of string = ('Choice A', 'Choice B', 'Choice C');
begin
    Result := false;
    bRA := 1;
    bRB := 1;
    bChap := GCOFF;
    bChbp := GCON;
    bChcp := GCOFF;
    bV := 10;

    // Define navigation strings
    cF := 'Nav:          ';
    cF[5] := CHUP;
    cF[6] := CHDN;
    cF[7] := CHLFT;
    cF[8] := CHRGT;
    cF[9] := CHTAB;
    cF[10] := CHESC;
    cF[11] := CHBTRGT;

    cI := 'Nav:    ^cS^cE';
    cI[5] := CHLFT;
    cI[6] := CHRGT;
    cI[7] := CHESC;
    cI[8] := CHBTRGT;

    cR := 'Nav:          ';
    cR[5] := CHUP;
    cR[6] := CHDN;
    cR[7] := CHLFT;
    cR[8] := CHRGT;
    cR[9] := CHTAB;
    cR[10] := CHESC;
    cR[12] := CHBTRGT;

    cX := 'Nav:X         ';
    cX[7] := CHTAB;
    cX[8] := CHESC;
    cX[9] := CHBTRGT;

    // Define input string defaults
    cA := '-100.00                                 ';
    cB := 'This string has something to edit in it!';
    cC := '                                        ';
    cD := ' Any character string!                  ';
    cD[1] := CHBALL;
    cD[23] := CHBALL;

    // Set radio button and spinner previous selection defaults
    bRAp := bRA;
    bRBp := bRB;
    bVp := bV;

    // Open window & draw form
    bW1 := WOpen(2, 4, 36, 18, WOFF);
    WOrn(bW1, WPTOP, WPLFT, 'Input Form');
    WOrn(bW1, WPTOP, WPRGT, 'Edit');
    WOrn(bW1, WPBOT, WPLFT, cF);

    WPrint(bW1, 1, 1, WOFF, 'Data Fields');
    WPrint(bW1, 2, 2, WOFF, 'Numer:');
    WPrint(bW1, 2, 3, WOFF, 'Alpha:');
    WPrint(bW1, 2, 4, WOFF, 'AlNum:');
    WPrint(bW1, 2, 5, WOFF, 'Any..:');
    WPrint(bW1, 2, 6, WOFF, 'Spin.:');
    GSpin(bW1, 8, 6, 0, 100, bVp, GDISP);

    WPrint(bW1, 1, 8, WOFF, 'Radio Buttons (h)');
    GRadio(bW1, 2, 9, GHORZ, GDISP, bRAp, 3, prA);

    WPrint(bW1, 1, 11, WOFF, 'Radio Buttons (v)');
    GRadio(bW1, 2, 12, GVERT, GDISP, bRBp, 3, prB);

    WPrint(bW1, 20, 11, WOFF, 'Check Boxes');
    WPrint(bW1, 25, 12, WOFF, 'Milk');
    WPrint(bW1, 25, 13, WOFF, 'Bread');
    WPrint(bW1, 25, 14, WOFF, 'Butter');
    GCheck(bW1, 21, 12, GDISP, bChap);
    GCheck(bW1, 21, 13, GDISP, bChbp);
    GCheck(bW1, 21, 14, GDISP, bChcp);

    GButton(bW1, 21, 16, GDISP, 2, paB);

    // Display fields as is
    WPrint(bW1, 8, 2, WOFF, cA);
    WPrint(bW1, 8, 3, WOFF, cB);
    WPrint(bW1, 8, 4, WOFF, cC);
    WPrint(bW1, 8, 5, WOFF, cD);

    // Loop until form accepted
    repeat
        // ----- Display Input Fields -----
        // Show navigation info
        WOrn(bW1, WPBOT, WPLFT, cI);

        // Edit fields

        bA := GInput(bW1, 8, 2, GNUMER, 27, cA);
        bB := GInput(bW1, 8, 3, GALPHA, 27, cB);
        bC := GInput(bW1, 8, 4, GALNUM, 27, cC);
        bD := GInput(bW1, 8, 5, GANY, 27, cD);

        // ----- Spinner Input -----
        bV := GSpin(bW1, 8, 6, 0, 100, bVp, GEDIT);
        if (bV <> XESC) and (bV <> XTAB) then
        begin
            bVp := bV;
        end;

        // ----- Display Radio Buttons - horizontal -----
        // Show navigation info
        WOrn(bW1, WPBOT, WPLFT, cR);

        // Process buttons
        bRA := GRadio(bW1, 2, 9, GHORZ, GEDIT, bRAp, 3, prA);

        // If not bypass, set previous selected value
        if (bRA <> XESC) and (bRA <> XTAB) then
        begin
            bRAp := bRA;
        end;

        // Redisplay buttons
        GRadio(bW1, 2, 9, GHORZ, GDISP, bRAp, 3, prA);

        // ----- Display Radio Buttons - vertical -----
        bRB := GRadio(bW1, 2, 12, GVERT, GEDIT, bRBp, 3, prB);

        // If not bypass, set previous selected value
        if (bRB <> XESC) and (bRB <> XTAB) then
        begin
            bRBp := bRB;
        end;

        // Redisplay buttons
        GRadio(bW1, 2, 12, GVERT, GDISP, bRBp, 3, prB);

        // ----- Display Check Boxes -----
        // Set footer
        WOrn(bW1, WPBOT, WPLFT, cX);

        // Stay on this check until ESC, TAB, or set
        repeat
            // Display button and get choice
            bCha := GCheck(bW1, 21, 12, GEDIT, bChap);

            // If not ESC or TAB, set previous value
            if (bCha <> XESC) and (bCha <> XTAB) then
            begin
                bChap := bCha;
            end;
        until (bCha = XESC) or (bCha = XTAB);

        // Stay on this check until ESC, TAB, or set
        repeat
            // Display button and get choice
            bChb := GCheck(bW1, 21, 13, GEDIT, bChbp);

            // If not ESC or TAB, set previous value
            if (bChb <> XESC) and (bChb <> XTAB) then
            begin
                bChbp := bChb;
            end;
        until (bChb = XESC) or (bChb = XTAB);

        // Stay on this check until ESC, TAB, or set
        repeat
            // Display button and get choice
            bChc := GCheck(bW1, 21, 14, GEDIT, bChcp);

            // If not ESC or TAB, set previous value
            if (bChc <> XESC) and (bChc <> XTAB) then
            begin
                bChcp := bChc;
            end;
        until (bChc = XESC) or (bChc = XTAB);

        // Set footer
        WOrn(bW1, WPBOT, WPLFT, cF);

        // Prompt to accept form and redisplay buttons
        bM := GButton(bW1, 21, 16, GEDIT, 2, paB);
        GButton(bW1, 21, 16, GDISP, 2, paB);
    until bM <> XTAB;

    // Check for acceptance (OK button), and set exit flag
    if bM = 1 then
    begin
        Result := true;
        GAlert('Doing something with entered data...');
    end;

    // Close window
    WClose(bW1);
end;

// ------------------------------------------------------------
// Func...: ProgTest
// Desc...: Demos window status and progress bar.
// ------------------------------------------------------------
procedure ProgTest;
var
    bW1, bW2, bL, bS: Byte;
    iV: Word;
begin
    // Open status window
    bW1 := WOpen(9, 2, 20, 14, WOFF);
    WOrn(bW1, WPTOP, WPLFT, 'Status');
    WPrint(bW1, 1, 1, WOFF, 'Window Status');

    WPrint(bW1, 1, 2, WOFF, '------ ------');

    // Open progress bar window
    bW2 := WOpen(7, 18, 24, 4, WOFF);
    WPrint(bW2, 2, 1, WOFF, 'Progress:');

    // Display initial progress bar
    GProg(bW2, 2, 2, 0);

    // Loop through each window handle
    for bL := 0 to 9 do
    begin
        // Get the window status
        bS := WStat(bL);

        // Print the window handle #
        WPos(bW1, 6, 3 + bL);
        WPut(bW1, Char(bL + 48));

        // Print the window handle status
        if bs = WON then
        begin
            WPrint(bW1, 8, 3 + bL, WOFF, 'Used');
        end
        else begin
            WPrint(bW1, 8, 3 + bL, WOFF, 'Free');
        end;

        // Update progress bar
        iV := ((bL + 1) mod 10) * 10;
        if iV = 0 then
        begin
            iV := 100;
        end;
        GProg(bW2, 2, 2, iV);

        // Wait 1 second
        Delay(1000);
    end;

    GAlert(' Press a key to continue. ');

    // Close windows
    WClose(bW2);
    WClose(bW1);
end;

// ------------------------------------------------------------
// Func...: About
// Desc...: About Dialog
// ------------------------------------------------------------
procedure About;
var
    bW1: Byte;
begin

    // Show window
    bW1 := WOpen(1, 6, 38, 15, WOFF);
    WOrn(bW1, WPTOP, WPLFT, 'About');
    WPrint(bW1, WPCNT,  1, WOFF, 'Demo Application');
    WPrint(bW1, WPCNT,  2, WOFF, 'Version 1.00-PAS');
    WPrint(bW1, WPCNT,  3, WOFF, '(C) 2022  Wade Ripkowski, amarok');
    WPrint(bW1, WPCNT,  5, WOFF, 'Application to demonstrate');
    WPrint(bW1, WPCNT,  6, WOFF, 'the MadPascal library.');
    WPrint(bW1, WPCNT,  7, WOFF, 'Converted from C.');
    WPrint(bW1, 4,      9, WOFF, 'V1-2021-Atari8: Action!');
    WPrint(bW1, 2,     10, WOFF, 'V1-C-2022-Atari8: C (CC65)');
    WPrint(bW1, 2,     11, WOFF, 'V1-PAS-2022-Atari8: PAS (MadPascal)');
    WPrint(bW1, WPCNT, 13, WON,  ' Ok ');

    // Wait for key
    WaitKCX(WOFF);

    // Close window
    WClose(bW1);
end;

// ------------------------------------------------------------
// Func...: SubMenu
// Desc...: Sub menu routine
// ------------------------------------------------------------
procedure SubMenu;
var
    bW1, bC: Byte;
    bD: Boolean;
const
    pcM: array[0..2] of string = (' Sub-Item 1 ', ' Sub-Item 2 ', ' Sub-Item 3 ');

begin
    bD := false;

    // Open window
    bW1 := WOpen(16, 10, 14, 5, WOFF);
    WOrn(bW1, WPTOP, WPLFT, 'Sub-Menu');

    // Loop until exit
    while not bD do
    begin
        // Display menu and get choice
        bC := MenuV(bW1, 1, 1, WOFF, 1, 3, pcM);

        // Process choice
        case bC of
            1: GAlert(' Sub-Item 1 selected. ');
            2: GAlert(' Sub-Item 2 selected. ');
            3: GAlert(' Sub-Item 3 selected. ');
            XESC: bD := true;
        end;
    end;

    // Close window

    WClose(bW1);
end;

// Variables
var
    bW1, bW2, bC: Byte;
    bD: Boolean;
const
    pcM: array[0..4] of string =
      (' Input Form   ', ' Progress Bar ', ' Sub-Menu     ', ' About        ', ' Exit         ');
begin
    bD := false;

    // Setup screen
    WInit;
    WBack(14);

    // Open header window
    bW1 := WOpen(0, 0, 40, 3, WON);
    WPrint(bW1, WPCNT, 1, WOFF, 'D E M O N S T R A T I O N');

    // Open menu window
    bW2 := WOpen(12, 7, 16, 9, WOFF);
    WOrn(bW2, WPTOP, WPCNT, 'Menu');

    // Loop until done (Exit selected)
    while not bD do
    begin
        // Call menu
        bC := MenuV(bW2, 1, 2, WON, 1, 5, pcM);

        // Process choice
        case bC of
            1: FormInput;
            2: ProgTest;
            3: SubMenu;
            4: About;
            5: bD := true;
        end;

        // Exit on ESC as well
        if bC = XESC then
        begin
            bD := true;
        end;
    end;

    // Close windows
    WClose(bW2);
    WClose(bW1);
end.

